# Maintainer: Andrea Dalle Vacche adallevacche@bloomberg.net

_realname=Carp-Clan
pkgname=perl-${_realname}
pkgver=6.08
pkgrel=1
pkgdesc="Carp::Clan - Report errors from perspective of caller of a \"clan\" of modules"
arch=('any')
license=('GPL' 'PerlArtistic')
url="https://search.cpan.org/~stbey/Carp-Clan/"
depends=('perl')
groups=('perl-modules')
options=('!emptydirs')
source=(https://cpan.metacpan.org/authors/id/E/ET/ETHER/${_realname}-${pkgver}.tar.gz)
sha256sums=('c75f92e34422cc5a65ab05d155842b701452434e9aefb649d6e2289c47ef6708')

build() {
  cd ${_realname}-${pkgver}
  perl Makefile.PL INSTALLDIRS=vendor
  make
}

check() {
  cd ${_realname}-${pkgver}
  make test
}

package() {
  cd ${_realname}-${pkgver}
  make DESTDIR="${pkgdir}" install
}
